
Senicpath1.sas 
/********************** senicpath1.sas **************************/ 

%include 'SenicRead.sas'; 

title2 'Big path analysis: Observed variables only'; 

 

proc calis cov;    /* Analyze the covariance matrix (Default is corr). */ 

     title3 'Full Model'; 

     var  stay age nbeds census nurses service /* Name the observed vars */ 

          mschool r1 r2 r3 infrisk culratio xratio;   

     /* Now give simultaneous equations, separated by commas. Latent 

        variables begin with F for factor. Error terms begin with 

        E for error or D for disturbance. SAS is not case sensitive. 

        You must name all the parameters. Optional starting values in  

        parentheses may be given after the parameters. */ 

     lineqs 

         infrisk = g1 stay + g2 age + g3 nbeds + g4 census + g5 nurses  

                   + g6 service + g7 mschool + g8 r1 + g9 r2 + g10 r3 + e1, 

          xratio   = b1 infrisk + e2, 

          culratio = b2 infrisk + e3;  

 

     std   /* Variances (not standard deviations) of exogenous vars  

              will be called phi-something. Colon means fill in the numbers. 

              Notice how we count the variances. Omitting the count will  

              generate warnings that say "Shorter parameter list than  

              variable list ..." These are harmless but unsettling.  

              Sometimes it is worth it not to have to count parameters, 

              but not this time. */ 

          stay age nbeds census nurses service mschool r1 r2 r3 = 10 * phi:, 

          e1 e2 e3 = 3 * psi: ;  /* And variances of error terms are psi. */   

     cov   /* Covariances: If not mentioned, it's zero.  Count the off- 

              diagonal elements. Call them kov, though they are really 

              phi_{i,j} for i not equal to j.  */ 

          stay age nbeds census nurses service mschool r1 r2 r3 = 45 * kov: , 

          e2 e3 = psi23; 

     bounds 0.0 < phi1-phi10 psi1-psi3;  /* Variances are positive */  


